Fehlende Textteile aus „Klein aber PICfein“ von Robert Schwager, erschienen in PCNEWSedu-49, „Embedded Systems“. Die Bilder sind hier eingebettet, die Größe der Bilder müßte man noch anpassen. 

PIC16C71

Der PIC16C71 bietet als Peripherie neben 4 analogen Eingängen bzw. 5 allgemeinen digitalen Ein/Ausgängen bzw. einen Timer/Zähler, 8-Bit TTL-Ein/Ausgänge und eine Interruptverarbeitung. Der Speicher, welcher das Programm beinhaltet, hat eine Größe von maximal 1024 Zeilen mit je einer Datenwortbreite von 14 Bits. 

� EINFÜGENGRAFIK C:\\pcn\\49\\pic\\abb_4.gif \* FORMATVERBINDEN ���

Abb. 4: Blockschaltbild des PIC16C71

Anschlußbelegung

� EINFÜGENGRAFIK C:\\pcn\\49\\pic\\abb_5.gif \* FORMATVERBINDEN ���

Abb. 5: Pinbelegung des PIC16C71

Programmspeicher und Stack

Der PIC16C71 hat einen 13-Bit-langen Programmzähler (PC - Program Counter), welcher einen 8Kx14 großen Programmspeicher verwalten kann. Jedoch sind nur die ersten 1Kx14 (0000h-03FFh) im Chip implementiert. Der Zugriff über dem Adreßbereich 3FFh verursacht einen „wrap-around“ (=„Ringsumher“) in dem ersten 1Kx14 Speicherraum. 

Der RESET-Vektor ist im Speicherplatz 0000h.

Der INTERRUPT-Vektor ist im Speicherplatz 0004h.

Der PIC16C71 hat 8 Stack-Speicherstellen für Unterprogrammroutinen (CALL, RETLW, RETFIE, RETURN).

Register-Organisation

00 indirect Addr. (8 bits) 		{= 80h}

01 TMR0 		(=RTCC)

02 PCL (PC Low order 8 bit) 	{= 82h}

03 STATUS 			{= 83h}

04 FSR 				{= 84h}

05 Port A (!) 			{85 TRIS Port A}

06 Port B 			{86 TRIS Port B}

08 ADCON0 (!)

09 ADRES (!)			{= 89h}

0A PCLATH (!)			{= 8Ah}

0B INTCON (!)			{= 8Bh}

0C..2F 36 General Purpose Registers (SRAM)

81 OPTION (!)	< Register Page auf 1 setzen! >

85 TRIS A	< Register Page auf 1 setzen! >

86 TRIS B	< Register Page auf 1 setzen! >

88 ADCON1 (!)	< Register Page auf 1 setzen! >



Achtung: Wenn eine Registeradresse, welche den Wert größer-gleich 80h besitzt, modifiziert werden soll, muß mittels PR0-Bit des STATUS-Register (03h) die ERSTE Seite der Register-Seiten durch ein Setzen dieses Bits ausgewählt werden. Man muß darauf achten, daß man dieses wieder zurücksetzt.

Das STATUS-Wort-Register

STATUS [03h] = (7) IRP|RP1|RP0|/TO|/PD|Z|DC|C (0)

Das STATUS-Byte beinhaltet die wichtigsten Operationen für ALU und Seitenadressierung.

C	Carry|/Borrow Bit	siehe PIC16C5X

DC	Digit Carry | /Borrow Bit	siehe PIC16C5X

Z	Zero Bit	siehe PIC16C5X

/PD	neg. Power Down Bit	siehe PIC16C5X

/TO	neg. Time Out Bit	siehe PIC16C5X



RP1, PR0: Register Page Select Bits for direct addressing

(Register-Seiten-Auswahl)

Mit diesen zwei Bits kann man zwischen 4 „Registerseiten“ für die Register auswählen, welche je 128 Bytes beinhaltet.

	RP1,0:	00	Seite 0 (000-07F)

		01	Seite 1 (080-0FF)

		10	Seite 2 (100-17F)

		11 	Seite 3 (180-1FF)



Jedoch nur das RP0 kann benutzt werden, das zweite Bit RP1 kann als setz- und löschbares Registerbit verwendet werden, da konkret der maximale verfügbare Register auf der Adresse 8Bh steht.

IRP: Register Page Select Bit for inderect adressing

Dieses Bit wird NICHT in dem PIC16C71 benutzt und kann als generelles Register-Bit verwendet werden.

Ein-Ausgabe Port A & B

PortA [05h] = (7) X | X |X | PA4,T0CKI (4)

(3) PA3,AIN3,VREF|PA2,AIN2|PA1,AIN1|PA0,AIN0 (0)

PortB [06h] = (7) PB7|PB6|PB5|PB4|PB3|PB2|PB1|PB0 (0)



TRISA [85h] und TRISB [86h]: Register für die Richtungsfestlegung der Datenrichtung je Bitstelle (0=Output, 1=Input)

Bemerkung: Der Befehl TRIS f ist zwar nicht im „regulärem" Befehlssatz vorhanden

Port A:

Der 5 Bit breite Ein/Ausgabeport hat 3 Funktionen, die man durch die Konfigrationsregister (STATUS, TRISA & ADCON1) einstellen kann.

digitale Ein/Ausgabe (max. 5)

analoge Eingabe (max. 4)

Zählereingang



Port B:

Die vier oberen Bits des Portes B (PB4..PB7) haben je einen ein/auschaltbaren Pullup-Widerstand, bei welchen ca. 250µA (typisch) fließt. Dieser Pull-Up-Widerstand wird bei einer Ausgabebeschaltung automatisch ausgeschaltet.

Registerport-B-Interrupt-Flag von PortB Bit 4 bis 7:

Tritt eine Pegelzustandsänderung auf eines der 4 Bits der 4 oberen Bits des Ports B (PB4 bis PB7) auf, so wird das RBIF-Bit im INTCON-Register (0Bh) Bit 0 gesetzt.

Analog-Digital-Wandler Konfigurationsregister ADCON0,1

ADCON0 [08h] = (7) ADCS1 | ADCS0 | - | CHS1 (4)

(3) CHS0 | DO,DONE |ADIF | ADON (0)

ADCON1 [88h] = (7) - | - | - | - | - | - | PCFG1 | PCFG0 (0)



ADON: A/D Off-oN

Ein/Ausschalten des Analog-Digital-Konverters

	ADON:	0	Ausschalten

		1	Einschalten



ADIF: A/D Interupt Flag

Nach Beendigung einer Konversion wird dieses Bit auf 1 gesetzt, jedoch sollte die Software nach Beendigung der Verarbeitung der AD-Wertes dieses Bit wieder löschen.

GO,/DONE: A/D Start Bit 

Dieses Bit muß gesetzt werden um ein Umwandlung zu beginnen. Es wird automatisch gelöscht nach einer Umwandlung.

CHS1,CHS0: CHannel Select

Mit den zwei Bits kann man die Eingänge zu dem Analog-Digital-Konverter auswählen.

	CHS1,0:	00	Kanal 0 (AIN0)

		01	Kanal 1 (AIN1)

		10	Kanal 2 (AIN2)

		11	Kanal 3 (AIN3)



ADCS1,ADCS0: A/D conversion Clock Select

Mit den zwei Bits kann man die Taktrate des Analog-Digital-Konverters einstellen.

	ADCS:	00	fOSC/2

		01	fOSC/8

		10	fOSC/32

		11	fRC, Interner RC-Generator 



PCFG1,PCFG0: Port ConFiG

Mit diesen zwei Bits kann man die Ein-Ausgänge des Portes A je nach Verwendung - sei es als analoger Eingang,  Referenzspannungseingang oder bidirektionaler digitaler Port - konfigurieren.

PCFG1,0: 00	PA0,1,2 & 3	analoger Eingang

		Referenzspannung VDD

	   01	PA0,1 & 2	analoger Eingang

		Referenzspannung PA3

	   10	PA0 & PA1	analoger Eingang

		PA2 & PA3	digitaler Ein/Ausgang

		Referenzspannung VDD

	   11	PA0, 1, 2 & 3	analoger Eingang

		Referenzspannung VDD



AD-RESult register

ADRES [09h] = 8-Bit A/D result register

Dieses Register enthält das 8 Bit lange Ergebnis des A/D-Wandlers. UIN= URef * ADRES / 256

PCLATH

PCLATH [0A] = (7) holding register for high byte of PC (0) Höheres Bytes des Programmzähler bzw. Programcounters.

INTCON

INTCON [0B] = 	(7) GIE | ADIE | T0IE | INTE  (4)

	(3) RBIE | T0IF |INTF | RBIF (0)

Mit diesem Byte kann man die Interruptverwaltung steuern.

RBIF: Registerport B Interrupt Flag

Dieses Bit wird gesetzt, wenn an Port B Bit 4 bit 7 ein Pegelwechsel stattfindet. Das Bit muß softwaremäßig zurückgesetzt werden. (siehe Ein-Ausgabe Port A & B Registerport-B-Interrupt-Flag von PortB Bit 4 bis 7)

INTF: INTerrupt Flag

Dieses Bit wird gesetzt, wenn über Port B BitNr. 0 ein Interrupt anliegt. Das Bit muß softwaremäßig zurückgesetzt werden. Das Bit muß softwaremäßig zurückgesetzt werden.

T0IF: Tmr0 Interrupt Flag

Dieses Bit wird gesetzt, wenn ein Zählerüberlauf (FFh->00h) (Register TMR0 01h) stattgefunden hat. Das Bit muß softwaremäßig zurückgesetzt werden.

RBIE: Registerport B Interrupt Enable bit

Mit dem Setzen dieses Bits wird der Aufruf der Interruptserviceroutine bei dem Interrupt RBIF, welcher durch eine Pegeländerung an PB4 bis PB7 ausgelöst wird, ermöglicht, jedoch nur wenn GIE gesetzt ist.

INTE: INTerrupt Enable bit

Mit dem Setzen des Bits wird bei einem Auftreten eines Interruptsignales an PB0 ein Aufrufen des Interruptserviceprogramms ermöglicht, jedoch nur wenn GIE gesetzt ist.

T0IE: Tmr0 Interrupt Enable

Mit dem Setzen dieses Bits wird der Aufruf des Interrupt�serviceprogramms bei einem Zählerüberlauf (FFh->00h) des RTCC (01h), jedoch nur wenn GIE gesetzt ist, ermöglicht.

ADIE: A/D Interrupt Enable bit

Mit dem Setzten dieses Bits wird ermöglicht, daß eine Interruptserviceroutine nach einer A/D-Umwandlung ausgeführt wird, jedoch nur wenn GIE gesetzt ist.

GIE: Global Interrupt Enable

Mit dem Setzten wird generell bei Auftreten eines Interrupt und deren Freigabe (d.h. einschalten) das Interruptserviceprogramm ausgeführt.

Eine Formel soll dieses noch einmal veranschaulichen:

Aufruf der Interruptserviceroutine = GIE & [(RBIE & RBIE) + (INTE & INTF) + (T0IE & T0IF) + (ADIE & ADIF)]



(Erläuterung: + .. log. ODER, & .. log. UND)



OPTION

OPTION [81] = 	(7) /RBPU | INTEDG | RTS | RTE (4)

	(3) PSA | PS2 | PS1 | PS0 (0)

Das OPTION-Register mit der Adressierung 81h ist ein les- und schreibbares Byte, welche verschiedene Kontrollbits für die Konfiguration des Vorteilers, externen Interrupt, dem TMR0 und dem „schwachen“ Pull-Up-Strom von typischen 250µA durch einen Pull-Up-Widerstand auf Port B beinhalten.

PS2,PS1,PS0 - PreScal Value

Vorzählereinstellung

	TMR0-Rate = 2PS[2:0]+1

	WDT-Rate = 2PS[2:0]



PSA - PreScale Assigment bit

Vorteilerzuweisung

	0 .. Anweisung von TMR0

	1 .. Anweisung von WDT



RTE - RTcc Signal Edge

Echtzeit-Zähler-Signal-Flankenerkennung

	0 .. Inkrement by LOW-HIGH-Flanke

	1 .. Inkrement by HIGH-LOW-Flanke



RTS - RTcc signal Source

Echtzeit-Zähl-Signal-Quelle

	0 .. Zähleingang = CLKOUT (¼ fOSZ)

	1 .. Zähleingang = TMR0-Eingang



INTEDG - INTerrupt Edge select

Auswahl der Flanken des Interruptsignales an PB0/INT

INT	0	Interrupt bei fallender Flanke

	1	Interrupt bei steigender Flanke



/RBPU - Registerport B Pull-Up enable

Ein gesetztes Bit schaltet die internen Pull-Up-Widerstände ab, welches auch bei einer Ausgabebeschaltung automatisch durchgeführt wird. Ist es gelöscht, werden die 4 höheren Bits über einen Widerstand auf die Betriebsspannung gelegt.

Interrupt-Verarbeitung

Es gibt 4 verschiedene Interruptquellen bei dem PIC16C71:

# Extrener Interrupt am Pin RB0/INT

# Interrupt bei Timer/Zähler-Überlauf des Timer0

# Interrupt bei Beendung der A/D-Umwandlung

# Interrupt bei Änderung des Signalpegels an PB4 bis PB7



Das Interrupt-Kontroll-Register (INTCON, Adresse 0Bh) zeichnet die einzelnen Interrupte in bestimmte Flags (T0IF, INF, RBIF) auf, außer das ADIF wird im ADCON (08h) aufgezeichnet, wenn diese auch nicht eingeschaltet sind.

Der globale Interrupt-Freigabe-Bit - GIE - (INTCON, Bit 7) soll mit einer 1 gesetzt werden, wenn alle „eingeschaltenen“ (unmaskierten) Interrupts das Interruptserviceprogramm aufrufen sollen. Das GIE-Flag wird bei einem RESET gelöscht.

Mit dem Befehl RETFIE kehrt der Prozeß aus der Interruptservceroutine zum unterbrochenen Programm zurück.

Wenn ein Interrupt ausgelöst wird, so wird das GIE-Bit für weitere Interruptsignale zunächst gelöscht, die Rücksprungadresse wird in den Stack-Speicher geschrieben, und der PC wird mit der Adresse 0004h geladen. Einmal in der Interruptserviceroutine eingelangt, können die Quelle(n) der Interrupt(s) durch Abfragen der 3 unteren Bits des Interrupt-Kontroll-Register (INTCON, Bit 0 bis 3) und des 2. Bits des ADCON erhalten werden. Die Interrupt-Flags (T0IF, INTF, RBIF, ADIF) müssen von der Software gelöscht werden, bevor man die Interrupts wiedereinschaltet, um rekursive Interrupts zu verhindern.

INT - Interrupt

Die Erkennung des externe Interrupts von Port B Bit 0 kann durch das INTEDG-Bit im INTCON-Register (0Bh) flankengesteuern werden. Ist INTEDG=1 so wird ein Interrupt bei steigender Flanke ausgelöst, d.h. von LOW (<0.2VDD) auf HIGH (<0.85VDD ). Hingegen INTEDG=0 ist dieses umgekehrt. Das Aufrufen des Interruptserviceprogramms bei Auftreten eines PB0-Interrupts kann mit einer 1 im INTE des INTCON-Registers (0Bh) eingeschaltet werden.

Ein Interrupt kann auch den Prozessor vom SLEEP-Befehl „aufgeweckt“ werden.

TMR0-Interrupt

Wenn ein Überlauf des TMR0 stattfindet, d.h. wenn das RTCC-Register von dem Wert FFh in den Wert 00h ändert, so wird das T0IF-Bit des INTCON (Bit 2) (0Bh) gesetzt. Dieser Interrupt kann durch das RBIE-Bit des INTCON Bit5 (0Bh) ein- (1) bzw. aus- (0) geschaltet werden.

A/D-Interrupt

Der A/D-Konverter setzt nach Beendigung der A/D-Umwandlung das Interrupt-Bit ADIF des ADCON Bit 1 (08h). Das Aufrufen des Interruptserviceprogramm nach Beendigung einer A/D-Konversion kann im ADIE-Bits des INTCON Bit 6 (0Bh) mit einer 1 ein- bzw. mit einer 0 ausgeschaltet werden.

Interruptbeispiel

; Programm: PRG3.ASM

LIST P=16C71

        

; Symbolzuweisung

        f        EQU 01h

        STATUS_R EQU 03h

        Port_A   EQU 05h

        Port_B   EQU 06h

        INTCON_R EQU 0Bh

        OPTION_R EQU 81h

        Tris_A   EQU 85h

        Tris_B   EQU 86h

        ADCON1_R EQU 88h

        

        #DEFINE Stop   Port_B,3

        #DEFINE RBIF   INTCON_R,0

        #DEFINE INTF   INTCON_R,1

        #DEFINE Page_1 STATUS_R,5



; Zuordnung des Resetvektors

ORG 000h

        GOTO Init



; Zuordnung des Interruptvektors

ORG 04h

        GOTO Int_Ser_Rou



; Programmzuweisung

ORG 010h



Init:   

        MOVLW b'10010000'

        MOVWF INTCON_R   ; Init. des INTCON-Reg.

        ; INT = 1 .. Interrupt Enable Bit EIN

        ; GIE = 1 .. Global Interrupt EIN

                

        BSF   Page_1

        ; Umschalten auf 1. Seite

        ; um OPTION-Register, Tris_A/B & ADCON zu

        ; konf. Der Assembler gibt einen Meldung

        ; (4x) aus, dass bei 8Xh-Adressierung das

        ; oberste Bit (also der 8-ter) wegge-

        ; schnitten wurde, dieses geschieht

        ; dadurch, dass bei der Register-

        ; adressierung nur 7 Bit (0-127)

        ; moeglich ist, daher muss auf die 

        ; 1. Seite umgeschaltet werden, nachher

        ; wieder auf Seite 0 zurueckschalten nicht 

        ; vergessen.



        MOVLW b'11000000'

        MOVWF OPTION_R    ; Init. des OPTION-Reg.

        ; /RBPU = 1 .. Pull Up AUS

        ; INTEG = 1 .. Int. bei steigender Flanke

        ; Auf Seite 1 schalten!!

               

        MOVLW b'11110111'

        MOVWF Tris_B      ; Init. d. IO-Dir-Reg.PB

        ; Port B 0,1,2,4,5,6 & 7 EINgaenge

        ; Port B 3 AUSgang

        ; Auf Seite 1 schalten!!



        MOVLW b'00000000'

        MOVWF Tris_A      ; Init. d. IO-Dir-Reg.PA

        ; Port A 0-4 AUSgaenge

        ; Auf Seite 1 schalten!!

        

        MOVLW b'00000011'

        MOVWF ADCON1_R    ; Init. des ADCON1-Reg.

        ; PA0-3 als digital E/A



        BCF   Page_1   ; Umschalten auf Seite 0



        BCF   Stop     ; Loeschen des Stop-Ausg.

        CLRF  Port_B   ; Loeschen des Port_B-Reg.



Prog:   INCF  Port_A,f ; Inkrementieren des PA's 

        GOTO  Prog     ; Sprungbefehl zu 'Prog'





Int_Ser_Rou:           ; INTerrupt SERvice ROUtine

        BSF   Stop     ; Setzen der Stop-Ausg.

                       ; Meldeausgang am PortB-3



Marke1: BTFSS RBIF     ; Ist RBIF=1, dann ueber-

                       ; springe naechsten Befehl

          GOTO Marke1  ; Sprung zu 'Marke1'



        BCF   RBIF     ; Loeschen des RBIF's 



Marke2: BTFSS RBIF     ; Ist RBIF=1, dann ueber-

                       ; springe naechsten Befehl

          GOTO Marke2  ; Sprung zu 'Marke2'

        

        BCF  Stop      ; Loeschen des Stop-Aug.

        BCF  INTF      ; Loeschen des INTF

        BCF   RBIF     ; Loeschen des RBIF's 



        RETFIE         ; Ruecksprungsbefehl 

                       ; des Interrupt-Service-R.



; Programmzuweisung bei Power On Reset

ORG 1FFh

        GOTO Init



; ------------------------------------------------

; Dieses Programm soll eine Verarbeitung einer 

; externen Interruptquelle an Bit 0 des Portes B 

; (PB0) demonstrieren. 

; Das Hauptprogramm des Demonstartionsprogramm

; ist hier kurz gehalten. Es wird das EA-Register

; PortA (PA0-PB4), welches als 5 Bit Ausgabeport 

; geschaltet ist, fortlaufend inkrementiert (um 

; 1 erhoeht), welche als Endlosschleife (mit einem

; GOTO-Befehl in die Zeile des INC-Befehls) 

; realisiert wurde.

; Die Interrupserviceroutine wird durch eine 

; steigenden Flanke an dem Pin PB0 ausgeloest.

; Der PC wird fuer einen RETFIE-Befehl gespeichert

; und nimmt den Wert PC=004h an, in welchem das

; Interrupt-Service-Programm steht. In diesem

; Programm steht ein GOTO-Befehl, natuerlich kann

; auch in und ab der Zeile 004h das Interrupt-

; Service-Routen-Programm stehen.

; Das hier implementiert Interruptservice-Programm

; haltet die Inkrementation an, da kein INC-Befehl

; im IntSer-Prg steht, und gibt an Port B Bit 3 

; (PB3) eine 1 aus, welches den Benuetzer anzeigen 

; soll, dass das System gestoppt hat, daher auch 

; der Name dieses Bits STOP.

; Um aus diesem Zustand zu gelangen, muessen 2

; Zustandaenderung an PortB Bit 4 bis 7 durchge-

; fuehrt werden. Wenn man einen PIC16C71 program-

; miert hat, kann man dieses mit einer Taste real-

; isieren (drucken und wieder loslassen; egal ob

; Schliesser oder Oeffner). Hier wird gezeicht, 

; dass zwar eine Interruptquelle kontrolliert 

; werden kann, ohne diese eine Interruptservice-

; routine aufzurufen, da das Ein/Ausschalt 

; geloescht ist.

; ! Man muss immer darauf achten, dass ein Interrupt

; erkennungsbit nach der Verarbeitung (meisten ein 

; Nachschauen) geloescht wird. Eine Interrupt-

; serviceroutine wird nur bei einer steigenden 

; Flanke eines eingeschlatenen Freigebebit der 

; zugehoere Interruptquelle aufgerufen.

; ! Man muss auch darauf achten, dass man wieder 

; das GIE-Bit setzt.

; Wenn man dieses Programm mit dem Simulator 

; testet kann man die Zustandaenderung der 

; Register gut nachvollziehen.

END

Programmlisting 3: Beispiel für INTERRUPT an PB0

Erweiterung bei TMR0 / WDT

Der TMR0-Modul Timer/Zähler hat folgende Merkmale:

8-Bit Timer/Zähler

Les- und schreibbares Register TMR0 (01h)

8-Bit softwareprogrammierbarer Vorteiler

interne bzw. externe Taktquelle

Interrupt bei einem Überlauf von FFh zu 00h

Flankenerkennung für Zählfunktion bei ext. Eingang

�Die Funktion des TMR0/WDT ist vergleichbar mit RTCC/WDT des PIC16C5x. 

Das foldene Programmbeispiel zeigt einen 16-Bit Zähler mit einem 8-Bit-Timer unter Verwendung der Interruptroutine.

; Programm: PRG4.ASM

LIST P=16C71

        

; Symbolzuweisung

        W        EQU 00h

        f        EQU 01h



        RTCC_L   EQU 01h

        RTCC_H   EQU 11h



        STATUS_R EQU 03h

        Port_A   EQU 05h

        Port_B   EQU 06h



        INTCON_R EQU 0Bh

        OPTION_R EQU 81h

        Tris_A   EQU 85h

        Tris_B   EQU 86h

        ADCON1_R EQU 88h

        

        #DEFINE Page_1 STATUS_R,5

        #DEFINE T0IF   INTCON_R,2



; Zuordnung des Resetvektors

ORG 000h

        Goto Init



; Zuordnung des Interruptvektors

ORG 04h

        INCF RTCC_H,f

        BCF  T0IF

        RETFIE

        



; Programmzuweisung

ORG 010h



Init:              

        MOVLW b'10100000'

        MOVWF INTCON_R   ; Init. des INTCON-Reg.

        ; T0IE = 1 .. Interrupt Enable Bit EIN

        ; GIE  = 1 .. Global Interrupt EIN

                

        BSF   Page_1

        ; Umschalten auf 1. Seite

        ; um OPTION-Register, Tris_A/B & ADCON zu

        ; konf. Der Assembler gibt einen Meldung

        ; (4x) aus, dass bei 8Xh-Adressierung das

        ; oberste Bit (also der 8-ter) wegge-

        ; schnitten wurde, dieses geschieht

        ; dadurch, dass bei der Register-

        ; adressierung nur 7 Bit (0-127)

        ; moeglich ist, daher muss auf die 

        ; 1. Seite umgeschaltet werden, nachher

        ; wieder auf Seite 0 zurueckschalten nicht 

        ; vergessen.



        MOVLW b'10000011'

        MOVWF OPTION_R    ; Init. des OPTION-Reg.

        ; /RBPU = 1 .. Pull Up AUS

        ; RTS = 0 Eingang = f osc/4

        ; RTE = 0 L-H-Flanke

        ; PSA = 0 Vorteilerzuweisung an f osc/4

        ; PS210 = 011 Vorzaehler-Rate = 16

        ; Auf Seite 1 schalten!!

               

        MOVLW b'00000000'

        MOVWF Tris_B      ; Init. d. IO-Dir-Reg.PB

        ; alle AUSgabe

        ; Auf Seite 1 schalten!!



        MOVLW b'00011111'

        MOVWF Tris_A      ; Init. d. IO-Dir-Reg.PA

        ; Port A 0-4 EINgaenge

        ; Auf Seite 1 schalten!!

        

        MOVLW b'00000011'

        MOVWF ADCON1_R    ; Init. des ADCON1-Reg.

        ; PA0-3 als digital E/A



        BCF   Page_1   ; Umschalten auf Seite 0



        CLRF  Port_B   ; Loeschen des Port_B-Reg.

        CLRF  RTCC_L   ; Loeschen des RTCC_L 

        CLRF  RTCC_H   ; Loeschen des RTCC_H



Prog:   SWAPF RTCC_L,W

        SUBWF RTCC_H,W

        MOVWF Port_B

        GOTO  Prog



; -----------------------------------------------

; Dieses Demoprogramm soll den Umgang mit dem

; Zaehlerinterrupt (T0IF) zeigen.

;   Es soll einen 16-Bit Zaehler mit Hilfe des

; Interrupts realsiert werden, wobei RTCC_L

; das eigentliche RTCC-Register (01h) und das

; RTCC_H das hoeherwertige Byte des Wortes ist,

; welche von einem Register (General Purpose

; Register) (Adresse 11h) verwendet wird.

;   Das Hauptprogramm ist eigentlich in diesem

; Fall unrelevant. Hier koennte jedoch eine Ver-

; arbeitung der zwei Register RTCC_H & RTCC_L

; stehen. Es sollte jedoch einen so hohen 

; Verteiler des RTCC und somit niedrige Zyklenzeit

; gewaehlt werden, so daá keine Aenderung der 

; beiden Register in der Verarbeitung stattfindet.

;   In diesem Programmbeispiel wurde fuer die 

; Simulation die RTCC-Quelle das Viertel des 

; Oszillatorfrequenz gewaehlt.

; .

; Simulator-Bericht:

; -PC- --- Befehl ---  T_H T_L -INTCON- STACK Bem

; 001F SWAPF RTCC_L,W  00h FFh 10100000  -- 

; 0020 SUBWF RTCC_H,W  00h FFh 10100000  --

; 0004 INCF  RTCC_H,f  00h 00h 00100100  21h   !

; 0005 BCF   T0IF      01h 00h 00100100  21h

; 0006 RETFIE          01h 00h 00100000  21h

; 0021 MOVWF Port_B    01h 00h 10100000  --

; 0022 GOTO  Prog      01h 00h 10100000  --

; Legende: PC = Program Counter

; T_H = RTCC_H (11h)   T_L = RTCC_L (01h)

END

Programmlisting 4: Beispiel für 16-Bit-Zähler mit INTERRUPT 

A-D-Konverter

Ein kleines „Kochrezept“ für die Programmierung der beiden  AD-Kontrollbytes sollten beachtet werden.

(1) Konfigurieren des ADCON1 [88h]

	Wählen der analogen/digitalen/Referenz-Eingänge

	(PCFG1,0) [Bit 1 & 0]

(2) Konfigurieren des ADCON0 [08h]

	Wählen der Taktrate des ADCs (ADCS1,0 - Bit7 & 6)

	Wählen der Eingangskanals (CHS1,0 - Bit 4 & 3)

	Einschalten des ADCs (ADON - Bit 0)

(3) Konfigurieren des A/D-Interrupts, wenn ein Interrupt im Programm benötigt wird.

	Löschen des ADIF (ADCON Bit 1)

	Setzen des ADIE (INTCON Bit 6)

	Setzen des globalen Interrupts GIE (INTCON Bit 7)

(4) Starten der Umwandlung

	Setzen desGO|/DONE-Bit (ADCON Bit2)



Beispiel:

Falsche Programmierung des ADCs:

	MOVLW B’11000111’

	MOVWF ADCON0



Richtiger Ablauf:

	MOVLW B’11000011’

	MOVWF ADCON0

	BSF   ADCON0,2

Hier sieht man, daß erst das ADCON0 ohne Start-Bit (GO|/DONE, 2. Bit des ADCON0-Registers (08h)) geladen werden muß, und dann erst dieses GO|/DONE-Bit extra gesetzt wird.

AD-Wandler Taktschema

Ein AD-Wandler operiert mit einem Taktsignal. Hier beim PIC16C71 kann man mit den zwei Bits ADCS1,0 des ADCON0 (08h) die Taktrate einstellen. Die Auswahl muß mit einer hardwaremäßigen Einschränkung ausgewählt werden, daß die Umwandlungszeit tAD größer als 2 Mikrosekunden ist, dies gilt vorallem bei ADCS1,0:00-10

Ein Umwandlung dauert 10 tAD. 

Der maximale Fehler beträgt ±1 (2) LSB.



; Programm: PRG5.ASM

LIST P=16C71

; Programmerklaerung:

;   Das Hauptprogramm soll hier sehr simpel sein,

;   da nur der Port_B staendig inkrementiert wird.

;   Das Interruptserviceprogramm, welches nur 

;   durch eine Beendigung einer A/D-Konverion

;   ausgeloest wird, ueberprueft das Ergebnis des

;   analogen Einganges an PA0 ob groeaaer als

;   den Wert 80h ist. Ist dies der Fall, so wird 

;   an dem digitale Ausgang PA3 'OVER' ein 

;   posetives Signal ausgegeben. Nur waehrend der 

;   Interruptserviceroutine wird ebenfalls ein 

;   posetives Signal an PA2 'ConEnd' ausgesendet.



; * Symbolzuweisung *

        W        EQU 00h

        f        EQU 01h



        STATUS_R EQU 03h

        Port_A   EQU 05h

        Port_B   EQU 06h

        ADCON0_R EQU 08h

        ADRES_R  EQU 09h

        INTCON_R EQU 0Bh



        OPTION_R EQU 81h

        Tris_A   EQU 85h

        Tris_B   EQU 86h

        ADCON1_R EQU 88h

                

        #DEFINE Carry  STATUS_R,0

        #DEFINE Page_1 STATUS_R,5

        #DEFINE ADIF   ADCON0_R,1

        #DEFINE GO     ADCON0_R,2

        #DEFINE ConEnd Port_A,2

        #DEFINE Over   Port_A,3



; * Zuordnung des Resetvektors *

ORG 0000h

        Goto Init



; * Zuordnung des Interruptvektors *

ORG 0004h 

        ; # Setzen des ConEnd-Bites an PA2

        BSF     ConEnd

        ; # ADC-Ergebnis auslesen

        MOVF    ADRES_R,W

        ; # Subtarktion 80h-W -> W

        SUBLW   80h

        ; # Vergleichen

        ; Carry=1, wenn W>80h

        ; Carry=0, wenn W=<80h

        ; Ist Carry=0, so wird 'Over' gesetzt.

        BTFSS   Carry

          BSF   Over

        ; Ist Carry=1, so wird 'Over' geloescht.

        BTFSC   Carry

          BCF   Over

        BCF     ConEnd

        BSF     GO

        BCF     ADIF

        RETFIE



; * Programmzuweisung *

ORG 0010h



Init:              

        ; # Initialisieren des Prozessors        

        BSF   Page_1

        ; Umschalten auf 1. Seite

        ; um OPTION-Register, Tris_A/B & ADCON zu

        ; konf. Der Assembler gibt einen Meldung

        ; (4x) aus, dass bei 8Xh-Adressierung das

        ; oberste Bit (also der 8-ter) wegge-

        ; schnitten wurde, dieses geschieht

        ; dadurch, dass bei der Register-

        ; adressierung nur 7 Bit (0-127)

        ; moeglich ist, daher muss auf die 

        ; 1. Seite umgeschaltet werden, nachher

        ; wieder auf Seite 0 zurueckschalten nicht 

        ; vergessen.



        MOVLW b'10000000'

        MOVWF OPTION_R    ; Init. des OPTION-Reg.

        ; /RBPU = 1 .. Pull Up AUS

        ; RTE = 0 L-H-Flanke

        ; Auf Seite 1 schalten!!

               

        MOVLW b'00000000'

        MOVWF Tris_B      ; Init. d. IO-Dir-Reg.PB

        ; alle AUSgabe

        ; Auf Seite 1 schalten!!



        MOVLW b'00000000'

        MOVWF Tris_A      ; Init. d. IO-Dir-Reg.PA

        ; Port A 0-4 AUSgaenge

        ; Auf Seite 1 schalten!!

        

        MOVLW b'00000010'

        MOVWF ADCON1_R    ; Init. des ADCON1-Reg.

        ; PA0-1 als analog Einaenge

        ; PA3-4 als digital E/A



        BCF   Page_1   ; Umschalten auf Seite 0



        MOVLW b'11000001'

        MOVWF ADCON0_R   ; Init. des ADCON0

        ; ADCS1,0: 11..ADC-Oszillaotor: inter. RC

        ; CHS1,0: 00..ADC-In = PA0

        ; ADON=1: ADC eingeschaltet



        MOVLW b'11000000'

        MOVWF INTCON_R   ; Init. des INTCON-Reg.

        ; ADIE = 1 .. A/D-Int. Enable Bit EIN

        ; GIE  = 1 .. Global Interrupt EIN



        CLRF    Port_B

        CLRF    Port_A



; * Hauptprogramm *

Prog:   

        INCF    Port_B,f

        ; Inkrementieren von Port B

        GOTO    Prog



END

Programmlisting 5: Beispiel für AD-Betrieb

Befehle

fehlende Befehle: OPTION ,  TRIS

Diese beiden fehlenden Befehle können durch den Transportbefehl an die Registeradresse ersetzt werden.

OPTION    => 	MOVWF 81h

TRIS f       =>	MOVWF 85h ; für Port A

	MOVWF 86h ; für Port B



Bemerkung: Statt dieser Befehle muß man von der 0. Seite auf die 1. Seite der Registerseiten umschalten, siehe dazu RP0 im STATUS-Register.

zusätzliche Befehle:

ADDLW k

(W + k) -> W

Addiert den Inhalt von W und der Konstanten k und speichert diesen in W ab.

Statusbits: C, CD, Z		Zykluszeit: 1

SUBWF k

(W - k) -> W

Subtrahiert den Inhalt von W von der Konstanten k und speichert diesen in W ab.

Statusbits: C, CD, Z		Zykluszeit: 1

RETFIE

TOS -> PC , 1 -> GIE

Rücksprungbefehl von einem Interrupt. Der Stack wurde geschrieben, und die letzte Speicheradresse (Top Of Stack - TOP) wird in den PC geladen. Die Interrupts werden durch das Bit GIE des INTCON [0Bh] wieder gesetzt.

Statusbits: keine			Zykluszeit: 2

RETURN

TOS -> PC

Rücksprung von einem Unterprogramm. Der Stack wird ausgelesen, und die letzte Speicheradresse (Top Of Stack - TOP) wird in den PC geladen.

Statusbits: keine			Zykluszeit: 2

Andere Eigenschaften des PIC16C71

Die anderen Eigenschaften des PIC16C71 ist dem des PIC16C5x ähnlich und sogar gleich, die hier nicht nochmals erläutert werden sollen.

PIC16C84

Die Merkmale des PIC16C84 sind folgende:

13 digitale Ein/Ausgabepins

4 Interruptquellen (PB0, Pegelwechsel an PB4-7, TMR0-Überlauf, EEPROM-Schreibbeendigung)

EEPROM-Programmspeicher (1024x14)

EEPROM-Datenspeicher (64x8)

36 allgemeine Register



Blockdiagramm

� EINFÜGENGRAFIK C:\\pcn\\49\\pic\\abb_6.gif \* FORMATVERBINDEN ���

Abb. 6: Blockschaltbild des PIC16C84

Anschlußbelegung

� EINFÜGENGRAFIK C:\\pcn\\49\\pic\\abb_7.gif \* FORMATVERBINDEN ���

Abb. 7: Pinbelegung des PIC16C84

Programmspeicher und Stack

Der PIC16C84 besitzt wie der PIC16C71 einen 13-Bit-breiten Programmzähler (Programcounter, PC), welcher einen EEPROM-Speicher von nur 1k Länge mit einer Breite von 14 Bit ansteuert. So wie bei dem PIC16C71 ist der Resetvektor des PIC16C84ers auf der Adresse 0000h und der Interruptverkor des PIC16C84ers auf der Adresse 0004h festgelegt. Der PIC16C81 (vergleichbar mit PIC16C71) hat ebenfalls 8 Speicherstellen für Rücksprungadressen (Stack-Levels).

Register-Organisation

00 indirect Addr. (8 bits) 	{= 80h}

01 TMR0

02 PCL (PC Low order 8 bit) 	{= 82h}

03 STATUS 	{= 83h}

04 FSR 	{= 84h}

05 Port A  	{85 TRIS Port A}

06 Port B (!)	{86 TRIS Port B}

08 EEDATA (!)

09 EEADR (!)	

0A PCLATH (!)	{= 8Ah}

0B INTCON (!)	{= 8Bh}

0C..2F 36 General Purpose Registers (SRAM)

81 OPTION (!)	< Register Page auf 1 setzen! >

85 TRIS A	< Register Page auf 1 setzen! >

86 TRIS B	< Register Page auf 1 setzen! >

88 EECON1 (!)	< Register Page auf 1 setzen! >

89 EECON2 (!)	< Register Page auf 1 setzen! >



Achtung: Wenn eine Registeradresse, welche den Wert größer-gleich 80h besitzt, modifiziert werden soll, muß mittels PR0-Bit des STATUS-Register (03h) die ERSTE Seite der Register-Seiten durch ein Setzen dieses Bits ausgewählt werden. Man muß darauf achten, daß man dieses wieder zurücksetzt.

Das STATUS-Wort-Register

STATUS [03h] = (7) IRP|RP1|RP0|/TO|/PD|Z|DC|C (0)

Das STATUS-Byte beinhaltet die wichtigsten Operationen für ALU und Seitenadressierung.

C	Carry|/Borrow Bit	siehe PIC16C5X

DC	Digit Carry | /Borrow Bit	siehe PIC16C5X

Z	Zero Bit	siehe PIC16C5X

/PD	neg. Power Down Bit	siehe PIC16C5X

/TO	neg. Time Out Bit	siehe PIC16C5X

RP1, PR0	Register Page Select Bits for direct addressing

		siehe PIC16C71

IRP	Register Page Select Bit for indirect adressing

		siehe PIC16C71



Ein-Ausgabe Port A & B

PortA [05h] = (7) X | X |X | PA4,RT|PA3|PA2|PA1|PA0 (0)

PortB [06h] = (7) PB7|PB6|PB5|PB4|PB3|PB2|PB1|PB0 (0)

TRISA [85h] und TRISB [86h]: Register für die Richtungsfestlegung der Datenrichtung je Bitstelle (0=Output, 1=Input)

Bemerkung: Der Befehl TRIS f ist zwar nicht mehr im „regulärem" Befehlssatz vorhanden

Port A:

Das 4. Bit der Portes A (PA4) ist neben einen regulären digitalen Ein/Ausgabepin auch der RT- bzw. T0CKI-Eingang.

Port B:

Die vier Bits des Portes B (PB4..PB7) haben je einen ein/auschaltbaren Pullup-Widerstand, bei welchen ca. 250µA (typisch) fließt. Dieser Pull-Up-Widerstand wird bei einer Ausgabebeschaltung automatisch ausgeschaltet.

Registerport-B-Interrupt-Flag von PortB Bit 4 bis 7:

Tritt eine Pegelzustandsänderung bei eines der 4 Bits der 4 oberen Bits des Ports B (PB4 bis PB7) auf, so wird das RBIF-Bit im INTCON-Register (0Bh) Bit 0 gesetzt.

interne EEPROM-Register-Verwaltung & Konfiguration

EEDATA, EEADR, EECON1,2

EEDATA [08h] = (7) d7|d6|d5|d4|d3|d2|d1|d0 (0)

EEADR [09h] = (7) a7|a6|a5|a4|a3|a2|a1|a0 (0)

EECON1 [88h] = (7) -|-|-|EEIF|WRERR|WREN|WR|RD (0)

EECON2 [89h] = „Schreibinitialisierungsregister“



RD: ReaD control bit

Dieses Bit muß man auf 1 setzen, wenn man den EEPROM lesen möchte. Es kann nur gesetzt werden, da es durch die Hardware nach einem Zyklus, dem Lesezyklus gelöscht wird.

WR: Write control bit

Dieses Bit muß man auf 1 setzen, wenn man den EEPROM schreiben möchte. Es kann nur gesetzt werden, da es durch die Hardware nach einem Zyklus (dem Schreibezyklus) gelöscht wird.

WREN: Write Enable bit

Dieses Bit kann man als Schreibschutzbit angesehen werden. Ist dieses Bit mit einer 0 gekennzeichnet (WREN=0), so kann das EEPROM NICHT geschrieben werden. Ist dieses jedoch 1, so kann es beschrieben werden.

WRERR: WRite ERRor bit

Dieses Bit wird gesetzt, wenn ein Schreibbefehl nicht durchgeführt werden kann/könnte. Dieses kann durch einen /MCLR-Reset oder durch eine WDT-Auszeit verursacht werden.

EEIF: EEprom Interrupt Flag

Dieses Bit wird nach einem komplett durchgeführten Schreiben in das EEPROM gesetzt. Gelöscht wird dieses Bit per Software.

Wird das EEIE-Bit des INTCON Bit 2 gesetzt (und GIE=1), so wird das Interruptserviceprogramm aufgerufen.

PCLATH

PCLATH [0A] = (7) holding register for high byte of PC (0)

Höheres Bytes des Programmzählers bzw. Programcounters.

INTCON

INTCON [0B] = 	(7) GIE | EEIE | T0IE | INTE  (4)

	(3) RBIE | T0IF |INTF | RBIF (0)

Mit diesem Byte kann man die Interruptverwaltung steuern.

RBIF: Registerport B Interrupt Flag

�Dieses Bit wird gesetzt, wenn an Port B Bit 4 bit 7 ein Pegelwechsel stattfindet. Das Bit muß softwaremäßig zurückgesetzt werden. (siehe Ein-Ausgabe Port A & B Registerport-B-Interrupt-Flag von PortB Bit 4 bis 7)



INTF: INTerrupt Flag

Dieses Bit wird gesetzt, wenn über Port B BitNr. 0 ein Interrupt anliegt. Das Bit muß softwaremäßig zurückgesetzt werden.

T0IF: Tmr0 Interrupr Flag

Dieses Bit wird gesetzt, wenn ein Zählerüberlauf (FFh->00h) (Register TMR0 01h) stattgefunden hat. Das Bit muß softwaremäßig zurückgesetzt werden.

RBIE: Registerport B Interrupt Enable bit

Mit dem Setzen dieses Bits wird der Aufruf der Interruptserviceroutine bei dem Interrupt RBIF, welcher durch eine Pegeländerung an PB4 bis PB7 ausgelöst wird, ermöglicht, jedoch nur wenn GIE gesetzt ist.

INTE: INTerrupt Enable bit

Mit dem Setzen des Bits wird bei einem Auftreten eines Interruptsignales an PB0 ein Aufrufen des Interruptserviceprogrammes ermöglicht, jedoch nur, wenn GIE gesetzt ist.

T0IE: Tmr0 Interrupt Enable

Mit dem Setzen dieses Bits wird das Einschalten des Aufrufen des Interruptserviceprogrammes bei einem Zählerüberlaufes (FFh->00h) des RTCC (01h) ermöglicht, jedoch nur wenn GIE gesetzt ist.

EEIE: EEprom Interrupt Flag

Mit dem Setzten dieses Bits wird ermöglicht, daß eine Interruptserviceroutine nach einem Schreibzyklus in den EEPROM durchgeführt werden kann.

GIE: Global Interrupt Enable

Mit dem Setzten wird generell bei Auftreten eines Interrupts und deren Freigabe (d.h. einschalten) das Interruptserviceprogramm ausgeführt.

Eine Formel soll dieses noch einmal veranschaulichen:

Aufruf der Interruptserviceroutine = GIE & [(RBIE & RBIE) + (INTE & INTF) + (T0IE & T0IF) + (EEIE & EEIF)]

(Erläuterung: + .. log. ODER, & .. log. UND)

OPTION

OPTION [81] = 	(7) /RBPU | INTEDG | RTS | RTE (4)

	(3) PSA | PS2 | PS1 | PS0 (0)

Das OPTION-Register mit der Adressierung 81h ist ein les- und schreibbares Byte, welche verschiedene Kontrollbits für die Konfiguration des Vorteilers, externen Interrupt, dem TMR0 und dem „schwachen“ Pull-Up-Strom von typischen 250µA durch einen Pull-Up-Widerstand auf Port B beinhalten.

PS2,PS1,PS0 - PreScal Value

Vorzählereinstellung

	TMR0-Rate = 2PS[2:0]+1

	WDT-Rate = 2PS[2:0]



PSA - PreScale Assigment bit

Vorteilerzuweisung

	0 .. Anweisung von RTCC

	1 .. Anweisung von WDT



RTE - RTcc Signal Edge

Echtzeit-Zähler-Signal-Flankenerkennung

	0 .. Inkrement by LOW-HIGH-Flanke

	1 .. Inkrement by HIGH-LOW-Flanke



RTS - RTcc signal Source

Echtzeit-Zähl-Signal-Quelle

	0 .. Zähleingang = CLKOUT (¼ fOSZ)

	1 .. Zähleingang = TMR0-Eingang



INTEDG - INTerrupt EdGe select

Auswahl der Flanken des Interruptsignales an PB0/INT

INTEDG	0	Interrupt bei fallender Flanke

	1	Interrupt bei steigender Flanke



/RBPU - Registerport B Pull-Up enable

Ein gesetztes Bit schaltet die internen Pull-Up-Widerstände ab, welches auch bei einer Ausgabebeschaltung automatisch durchgeführt wird. Ist es gelöscht, werden die 4 höheren Bits über einen Widerstand auf die Betriebsspannung gelegt.

EEPROM-Register, EEPROM-Datenspeicher

Der PIC16C84 hat einen 64x8 EEPROM-Datenspeicher, welcher schreib- und lesbar ist. Dieser Speicher behält auch die Daten, wenn die Versorgungsspannung unterschritten worden ist. Der EEPROM wird durch die zwei Register EEDATA (08h) und EEADR (09h) unter Mitwirkung der beiden Initialisierungsregister EECON1 (88h) und EECON2 (89h) angesteuert. EEDATA beinhaltet die Daten für das Schreiben bzw.  vom Lesen des EEPROMs, welche durch die Adresse des EEPROM, welche im EEADR steht, hier bitte achten, daß nur 64 Speicherstellen (00h-3Fh) zur Verfügung stehen, „ausgewählt“. Das Initialisierungsregister EECON1 ist für die Art des Zugriffes des Speichers - LESEN oder SCHREIBEN - , dem Schreibschutz zuständig.

Hingegen ist das Register EECON2 nur für das „Initialisieren“ eines  Schreibens nützlich.

LESEN aus dem EEPROM-Datenspeicher

Das Lesen ist nach einem Rezept leicht nachvollziehrbar:

Schreiben des EEADR-Registers

Setzen des Read-RD-Bits des EECON1-Register (Bit 0)

Lesen der Daten der ausgewählten Speicherstelle zB. durch einen Transportbefehl.



; Programm: PRG6a.ASM

LIST P=16C84

; Lesen aus dem EEPROM-Speicher

; Dieses Programm ist ideal zur Demonstartion

; fuer den Simulator, bei welchen zuerst die 

; EEPROM-Adresse 01h mit einem Wert (zB. FFh)

; gesetzt werden sollt.



; * Definieren von Symbolen *

        W        EQU 0

        f        EQU 1

        STATUS_R EQU 03h

        EEDATA_R EQU 08h

        EEADR_R  EQU 09h

        INTCON_R EQU 0Bh

        OPTION_R EQU 81h

        TRIS_A   EQU 85h

        TRIS_B   EQU 86h

        EECON1_R EQU 88h

        EECON2_R EQU 89h



        #DEFINE Page_1 STATUS_R,5

        #DEFINE RD     EECON1_R,0





ORG 0000h

        GOTO Start



ORG 01FFh

        GOTO Start



ORG 0010h

Start:

        ; # Initialisieren des Prozessors

        CLRF    INTCON_R

                ; keine Interruptverarbeitung

        BSF     Page_1

        CLRF    TRIS_A    

                ; Port_A als AUSgabe

        CLRF    TRIS_B

                ; Port_B als AUSgabe

        CLRF    OPTION_R

                ; Timereingang auf fosc/4 ges.

		; kein Vorteiler

        BCF     Page_1



        ; # Setzen der EEPROM-Adresse mit 01h

        ;   Punkt 1

        MOVLW   01h

        MOVWF   EEADR_R



        ; # Setzen des READ-Bits im EECON1

        ;   Punkt 2

        BSF     Page_1

        BSF     RD

        BCF     Page_1



        ; # Auslesen der EEPROM-Daten (zB. in W)

        ;   Punkt 3

        MOVF    EEDATA_R,W



; * Endlosschleife

Marke:  NOP

        GOTO    Marke



END

Programmlisting 6a: Beispiel für EEPROM-LESEN

SCHREIBEN in den EEPROM-Datenspeicher

Setzen des WREN-Bits des EECON1-Register (Bit 3)

Schreiben des EEADR-Register

Schreiben des EEDATA-Register

Initialisierungsprogramm�	(a) EECON2 mit dem Wert 55h setzen�	(b) EECON2 mit dem Wert AAh setzen

Setzen des Write-WR-Bits des EECON1-Register (Bit 1)

Warten bis EEPROM-Zeile geschrieben worden ist.�

Dieses kann auch ein Interrupt auslösen, wenn man EEIE=1 im INTCON Bit 6 setzt. Oder Abwarten bis EEIF im EECON1 Bit4 gesetzt worden wurde. Das Rücksetzten durch die Software nicht vergessen. (nur wenn nötig)

(*7) Löschen des WREN-Bits (nur wenn nötig)



; Programm: PRG6b.ASM

LIST P=16C84

; Schreiben in den EEPROM-Daten-Speicher

; Dieses Programm soll in den EEPROM-Datenspeicher

; mit der EEPROM-Adresse 01h den Wert 0Fh 

; 'einschreiben'. Der Prozess soll erst nach

; vollstaendigen Beendigung des Schreibens weiter-

; gefuehrt werden.



; * Symboldefinition *

        W EQU 0

        f EQU 1

        STATUS_R EQU 03h

        EEDATA_R EQU 08h

        EEADR_R  EQU 09h

        INTCON_R EQU 0Bh

        OPTION_R EQU 81h

        TRIS_A   EQU 85h

        TRIS_B   EQU 86h

        EECON1_R EQU 88h

        EECON2_R EQU 89h



        #DEFINE Page_1 STATUS_R,5

        #DEFINE WR EECON1_R,1

        #DEFINE WREN EECON1_R,2

        #DEFINE EEIF EECON1_R,4

        



ORG 0000h

        GOTO Start



ORG 01FFh

        GOTO Start



ORG 0010h

Start   

        ; # Initialisierung des Prozessors

        CLRF    INTCON_R

                ; keine Interruptverarbeitung

        BSF     Page_1

                ; Registerseite 1 !

        CLRF    TRIS_A    

                ; Port_A als AUSgabe

        CLRF    TRIS_B

                ; Port_B als AUSgabe

        CLRF    OPTION_R

                ; Timereingang auf fosc/4

                ; kein Vorteiler

        CLRF    EEDATA_R

                ; Loeschen des EEDATA-Registers

                ; (ideal f�r Simulation)

        CLRF    EEADR_R

                ; Loeschen des EEADR-Registers

                ; (ideal f�r Simulation)

        CLRF    EECON1_R

                ; Loeschen des EECON1-Register

                ; (ideal f�r Simulation)

        CLRF    EECON2_R

                ; Loeschen des EECON2-Register

                ; (ideal f�r Simulation)

        BCF     Page_1

                ; Registerseite 0



        ; # Sezten des WREN-Bits in EECON1-Bit3

        ;   Punkt 1

        BSF     Page_1

        BSF     WREN

        BCF     Page_1



        ; # Schreiben des EEADR-Register

        ;   Punkt 2

        MOVLW   01h

        MOVWF   EEADR_R

        

        ; # Schreiben des EEDATA-Register

        ;   Punkt 3

        MOVLW   0Fh

        MOVWF   EEDATA_R



        ; # Initialisierungsprogramm

        ;   Punkt 4

        BSF     Page_1

        MOVLW   55h

        MOVWF   EECON2_R

                ; EECON2 mit Wert 055h schreiben

                ; Punkt 4 a

        MOVLW   0AAh

        MOVWF   EECON2_R

                ; EECON2 mit Wert 0AAh schreiben

                ; Punkt 4 b

        ; # Setzen des WRITE-Bits im EECON1

        BSF     WR

        BCF     Page_1



        ; # Warteschleife bis Datum in den EEPROM

        ;   geschrieben ist. Dieses wird mit 

        ;   Setzen des Interrupt-Flags EEIF

        ;   im EECON1 Bit 4 angezeigt. 

        ;   Softwaremaessiges Loeschen 

        ;   erforderlich!

        ;   Punkt 6

        BSF     Page_1

Schleif BTFSS   EEIF

          GOTO  Schleif

        BCF     EEIF

        BCF     Page_1



; * Endlosschleife

Marke   NOP

        GOTO    Marke



END

Programmlisting 6b: Beispiel für EEPROM-SCHREIBEN



Das folgende Beispielprogramm soll eine Schreiben und ein Lesen des EEPROM-Daten-Speichers zeigen. 



; Programm: PRG6.ASM

LIST P=16C84



; * Symbolzuweisungen *

        W        EQU 0

        f        EQU 1

        STATUS_R EQU 03h

        PORT_A   EQU 05h

        PORT_B   EQU 06h

        EEDATA_R EQU 08h

        EEADR_R  EQU 09h

        INTCON_R EQU 0Bh

        OPTION_R EQU 81h

        TRIS_A   EQU 85h

        TRIS_B   EQU 86h

        EECON1_R EQU 88h

        EECON2_R EQU 89h



        #DEFINE Zero     STATUS_R,2

        #DEFINE Page_1   STATUS_R,5

        #DEFINE RD       EECON1_R,0

        #DEFINE WR       EECON1_R,1

        #DEFINE WREN     EECON1_R,2

        #DEFINE EEIF     EECON1_R,4

        #DEFINE RD_Con   PORT_A,0

        ; Controllbit fuer das Lesen

        #DEFINE WR_Con   PORT_A,1

        ; Controllbit fuer das Schreiben

        #DEFINE Wait_Con PORT_A,2

        ; Controllbit fuer den Schreibenzyklus

        ; im EEPROM des Datums



; * Definieren des Resetvektors *

ORG 0000h

        GOTO Start



; * Definieren des Startvektors *

ORG 01FFh

        GOTO Start



; * Definieren des Hauptprogrammadresse *

ORG 0010h



; * Hauptprogramm *

Start:  

; * Initialiesiern der Register *

        CLRF    INTCON_R

                ; Alle Interrupts aus.

        BSF     Page_1   

        ; Umschalten auf Seite 1 der Registerseite

        CLRF    TRIS_A

                ; PortA alle auf AUSgabe

        CLRF    TRIS_B

                ; PortB alle auf AUSgabe

        CLRF    OPTION_R

                ; OPTION-Register

                ;  Pull-Up ein (wirkungslos, 

                ;       da Port auf Ausgabe)

                ;  Timereinagng: CLKOUT (fosc/4)

                ;  Vorteiler: aus

        CLRF    EECON1_R

                ; Loeschen des Inhaltes in EECON1

        CLRF    EECON2_R

                ; Loeschen des Inhaltes in EECON2

        BCF     Page_1

                ; Zurueckschalten von Seite 1 auf 

                ;     Seite 0 der Registerseiten

        CLRF    EEDATA_R

                ; Loeschen des Inhaltes in EEDATA

        CLRF    EEADR_R

                ; Loeschen des Inhaltes in EEADR



        

; * SCHREIBEN der EEPROM mit Daten *

; Kurzbeschreibung:

;   Das Programm soll den ganzen EEPROM-

;   Speicher (00h-3Fh, 64 Bytes) mit dem

;   Wert der Summe aus dem Wert der Adresse mit

;   der Konstante 64d bzw. 40h beschreiben.

;       [EEDATA]=[EEADR]+40h

;   PortB dient als Zaehlermonitor der EEPROM-

;   Adressierung. Waehrend dem ganzen Beschreiben 

;   wird an PortA Bit 1 ein Write-Controllbit

;   - WR_Con - ausgegeben. Und bei dem 

;   Beschreiben und Abwarten der Beendigung des 

;   EEPROM-Schreibezyklus wird ein weiteres 

;   Meldebit Wait_Controllbit - Wait_Con - an

;   PortA Bit 2 ausgegeben.



; - Initalisieren der Ports

        CLRF    PORT_B

        CLRF    PORT_A

        BSF     WR_Con



; - Schreiberoutinen

Write:  

        ; # Setzen des WREN-Bit im EECON1 (Seite 1)

        BSF     Page_1

        BSF     WREN

        BCF     Page_1



        ; # Inhalt von PortB in W und in EEADR

        MOVF    PORT_B,W

        MOVWF   EEADR_R

               

        ; # Inhalt von PortB in W und Addition

        ;   des Wertes 64d, speichern in EEDATA

        MOVF    PORT_B,W

        ADDLW   d'64'

        MOVWF   EEDATA_R



        ; # EEPROM-Schreibsequenz !!!

        BSF     Page_1   

                ; Umschalten auf Seite 1 (Reg.)

        MOVLW   55h

        MOVWF   EECON2_R 

                ; EECON2=55h setzen

        MOVLW   0AAh

        MOVWF   EECON2_R 

                ; EECON2=0AAh setzen

        BSF     WR

                ; Setzen des WR-Bit im EECON1

        BCF     Page_1

                ; Zurueckschalten.



        ; # Kennzeichnung, dass EEPROM-Schreib-

        ;   Warte-Sequenz abgeschlossen ist,

        ;   mit setzen des Wait_Con.

        BSF     Wait_Con



        ; # Warteschelife bis EEPROM-Zeile

        ;   geschrieben worden ist mit der

        ;   Ueberpruefung des EEprom-Interrupt-Flags

        ;   ! Umschalten auf Seite 1 d. Reg.-S. !

        BSF     Page_1

Warten: BTFSS   EEIF

          GOTO  Warten



        ; # Loeschen des Wait_Con

        BCF     Wait_Con



        ; # Loeschen des EEprom-Interrupt-Flags

        BCF     EEIF



        ; # Zurueckschalten auf Seite 0

        BCF     Page_1



        ; # Zaehlsequenz von 00h bis 3Fh und 

        ;   Rueckkehr zu Schreibroutine

        ;   bei 40h austeigen.

        INCF    PORT_B,f

        MOVF    PORT_B,W

        SUBLW   d'64'

        BTFSS   Zero

          GOTO  Write



        ; # Umschalten des PortB auf Eingang.

        BSF     Page_1

        MOVLW   0FFh

        MOVWF   TRIS_A

        BCF     Page_1



; * LESEN der EEPROM mit Daten *

; Kurzbeschreibung:

;   In dieser Teilprogramm wird das Register fuer

;   die EEPROM-Adressierung (EEADR) von 00h bis

;   3Fh inkrementiert. Vor jeder Inkrementation 

;   wird der Inhalt dieser Adresse ausgelesen und

;   auf dem PortB ausgegeben. Bei dieser ganzen 

;   Aktion ist das RD-Controllbit (RD_Con) am 

;   PortA-Bit 0 (PA0) gleich 1. Nach Beendigung

;   der 64 Auslesezyklen des EEPROM verhart das 

;   System in eine Endlosschleife.



        ; # Loeschen des EEADR-Register, wegen

        ;   Zaehlerstandes!

        CLRF    EEADR_R



        ; # Setzen des RD_Controllbits an PA0

        BSF     RD_Con



; - Leseroutinen        

Read:   

        ; # Setzen des RD im EECON1

        ;   (Um & Zurueckschalten auf Seite 1)

        BSF     Page_1

        BSF     RD

        BCF     Page_1



        ; # Auslesen der Daten aus dem EEDATA-Reg.

        MOVF    EEDATA_R,W

        

        ; # Datzen auf PortB schreiben

        MOVWF   PORT_B     

        

; Bemerkung: Nach der Exekution des MOVWF PORT_B

;            Zeile ist der Inhalt des EEPROMs

;            am PortB (-> Simulator)



        ; # Zaehlsequenz von 00h bis 3Fh und 

        ;   Rueckkehr zu Leseroutine

        ;   bei 40h austeigen.

        INCF    EEADR_R,f

        MOVF    EEADR_R,W

        SUBLW   d'64'

        BTFSS   Zero

          GOTO  Read



        ; # Loeschen des PortA

        CLRF    PORT_A



; * Endlosschleife

Marke   NOP

        GOTO    Marke



; Bemerkung: Wenn man dieses Programm mittels

;   dem Simulator analysiert, sieht man genau

;   die Verarbeitung der Signale.

;            Es treten 13 Meldungen auf, dass

;   das hoechstwertige Bit ignoriert wird,

;   dieses hat zur Folge das eine Register-

;   adressierung ueber den Wert 80h nicht 

;   adressiert werden kann, da bei einer

;   Registeradressierung nur 7 Bit (statt 8)

;   zur Verfuegung stehen und daher auf die

;   erste Seite der Registerseitenadressierung

;   umgeschaltet werden muss.

END

Programmlisting 6: Beispiel für EEPROM

Interruptverarbeitung

Die Interruptverarbeitung des PIC16C84 ist der dem des PIC16C71 sehr ähnlich, jedoch besitzt der PIC16C84 keine AD-Wandler-Interruptverarbeitung, sondern eine für die Beendigung des EEPROM-Schreib-Zyklus.

andere Eigenschaften des PIC16C84

Die anderen Eigenschaften sind dem des PIC16C5x  ähnlich, für weitere Fragen steht natürlich das Datenbuch zur Verfügung.

Resümee PIC16C5x/71/84

PIC16C5x, PIC16C71 und PIC16C84 sind ideale Lern- und „Kleinsteuerungs“-Prozessoren. Die hier besprochenen Merkmale sind natürlich nur oberflächlich behandelt (für den Einstieg ideal). Für eine weitergehende Einarbeitung wird das Datenbuch empfohlen.








